# ###############################################################################
# #
# # Copyright (C) 2009 Jostein Bratlie
# # Contact: jostein.bratlie@gmail.com | http://bratlie.info
# #
# # This file is part of GMFlow, a Flow3D data visualizer
# #
# # GMFlow is free software: you can redistribute it and/or modify
# # it under the terms of the GNU Lesser General Public License as published by
# # the Free Software Foundation, either version 3 of the License, or
# # (at your option) any later version.
# #
# # GMFlow is distributed in the hope that it will be useful,
# # but WITHOUT ANY WARRANTY; without even the implied warranty of
# # MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the
# # GNU Lesser General Public License for more details.
# #
# # You should have received a copy of the GNU Lesser General Public License
# # along with GMFlow. If not, see <http://www.gnu.org/licenses/>.
# #
# ###############################################################################


## App
TARGET = MyGMapp
GMQTEXT_MODULES = glwidget messageboxes

## GMlib
DEFINES = GM_GL_EXTENSION

## Qt/qmake
TEMPLATE = app
CONFIG += qt debug_and_release static
QT += core gui opengl

## LIBS
include( gmqtext.pri )
include( gmlib.pri )

DEPENDPATH += src

## Platform spesifics
win32 { 
    contains( DEFINES, "GM_GL_EXTENSION" ):LIBS += -lGLee # GLee must be linked before the opengl libraries.
    LIBS += -lopengl32 -lglu32
}

unix {
    contains( DEFINES, "GM_SCRIPT_LUA" ):INCLUDEPATH += "/usr/include/lua5.1"
    contains( DEFINES, "GM_SCRIPT_LUA" ):LIBS += -llua5.1
}

## Paths
APP_ROOT = $$PWD
DESTDIR = $$APP_ROOT/bin
OBJECTS_DIR = $${OUT_PWD}/.obj
MOC_DIR = $${OUT_PWD}/.moc
RCC_DIR = $${OUT_PWD}/.rcc
UI_DIR = $${OUT_PWD}/.uic

## Files
FORMS +=
HEADERS += mainwindow.h \
    myglwidget.h \
    flow3ddata.h \
    flow3dnode.h \
    marchingcubes.h
SOURCES += main.cpp \
    mainwindow.cpp \
    myglwidget.cpp \
    flow3ddata.cpp \
    flow3dnode.cpp \
    marchingcubes.cpp
OTHER_FILES += \
    CHANGELOG
